#Maintainer: Jan de Groot <jgc@archlinux.org>

pkgname=xcursor-themes
pkgver=1.0.6
pkgrel=2
pkgdesc="X.org Cursor themes"
arch=(any)
url="https://xorg.freedesktop.org/"
license=('MIT')
makedepends=('pkgconfig' 'xorg-xcursorgen' 'xorg-util-macros')
source=(${url}/releases/individual/data/${pkgname}-${pkgver}.tar.bz2{,.sig})
sha512sums=('2ad3586a5571a5ddc0001187362b54b839644916164834213af98e58698044aec4392dd6e1e42f1794fda85c8fedf259b09214a9636bd0e71a1629c710f198d3'
            'SKIP')
validpgpkeys=('4A193C06D35E7C670FA4EF0BA2FB9E081F2D130E') # "Alan Coopersmith <alan.coopersmith@oracle.com>"

build() {
  cd ${pkgname}-${pkgver}
  ./configure --prefix=/usr
  make
}

package() {
  cd ${pkgname}-${pkgver}
  make DESTDIR="${pkgdir}" install
  install -m755 -d "${pkgdir}/usr/share/licenses/${pkgname}"
  install -m644 COPYING "${pkgdir}/usr/share/licenses/${pkgname}/"
}
